# Maintainer: Maxime Gauduin <alucryd@archlinux.org>
# Contributor: Tatsuyuki Ishi <ishitatsuyuki@gmail.com>

pkgname=python-blessed
pkgver=1.20.0
pkgrel=4
pkgdesc='A thin, practical wrapper around terminal styling, screen positioning, and keyboard input'
arch=(any)
url=https://github.com/jquast/blessed
license=(GPL)
depends=(
  python
  python-six
  python-wcwidth
)
makedepends=(
  git
  python-setuptools
)
checkdepends=(
  python-pytest
  python-pytest-cov
  python-pytest-rerunfailures
  python-pytest-xdist
)
_tag=d404d8ccde9ba900bc968c1213b915a691581603
source=(git+https://github.com/jquast/blessed.git#tag=${_tag})
b2sums=(SKIP)

pkgver() {
  cd blessed
  git describe --tags
}

build() {
  cd blessed
  python setup.py build
}

check() {
  cd blessed
  pytest
}

package() {
  cd blessed
  python setup.py install --root="${pkgdir}" --optimize=1 --skip-build
}

# vim: ts=2 sw=2 et:
